New Rules for Assessing Mathematical Competencies

Calculus with GeoGebra

Co-funded by the Erasmus+ Programme of the European Union

http://rules-math.com
@rules_math_proj
https://www.facebook.com/rulesmath
Project: 2017-1-ES01-KA203-038491

Contents

1 INTRODUCTION

The goal of this document is to present a guide for the most commonly used functionalities of GeoGebra in the scope of a Calculus subject. GeoGebra is a free application that can be either accessed online or installed at the computer, and that can be used for learning Calculus, Geometry, Algebra, and Statistics.

Along this document, instructions are numbered so it is easier to locate and follow them. Besides, the symbol “>” represents text that must be entered in the corresponding input box by students.

2 PRELIMINARIES

1) Go to "www.geogebra.org" and select English as the language at the bottom left corner of that web page (next to the globe icon).

2) Click on the link "GeoGebra Classic" under "Classic Apps" off-centre at the bottom of that web page.

3) GeoGebra Classic's options are accesible through the drop-down menu located at the upper right corner of its web page (the three-lined icon next to the magnifying glass icon). For example, we can save/load files, change again the language or select the elements (views) that will appear on screen.

4) At the drop-down menu, enable the options "CAS" and "Input Bar". Later during this session we will also use "3D Graphics", but it will not be necessary at the moment.

"Algebra" and "CAS" are similar, but there are some important differences: the elements entered in the Input Bar appear only in the "Algebra" view, while some commands only work in the "CAS" view (they need to be typed directly in the text boxes of the "CAS" view).

Numbers *e* and π can be entered by using Alt+E and Alt+P, or simply by writing e and pi, respectively.

3 FUNCTIONS

Functions are entered in the usual way, independently of the number of variables. If no identifier is given (e.g. `f(x)`), GeoGebra chooses the identifier for us.

1) Enter the following elements one by one through the Input Bar for the sine, cosine, tangent, exponential, absolute value, square root and cubic root functions:

```
> f_1(x)=sin(x)

> f_2(x)=cos(x)

> f_3(x)=tan(x)

> f_4(x)=e^(x)

> f_5(x)=abs(x)

> f_6(x)=sqrt(x)

> f_7(x)=x^(1/3)
```

Whenever there are several functions in the "Algebra" view, we can select which ones are visualized by clicking on the big, colored dot associated to each function.

2) Disable and enable the visualization of several functions.

We can change the colour of each function in two ways: either by right clicking the figure of the function and selecting "Settings" and the selecting the option "Colour" or by left clicking the icon composed of three icons located to the right of the function at the "Algebra" view and then selecting "Settings" and "Colour" (that menu also allows to delete the element). We can change the line thickness using the same procedure, but instead of the option "Colour" we must use the option "Style". It is also possible to enable/disable the label associated to the function.

3) Change the colour and thickness of some of the previous functions and enable/disable its label.

In the "Graphics" view, we can move the graphic of the funcion mantaining the left button of the mouse pressed and then moving the mouse. Besides, we can zoom in or zoom out using the two magnifying glass icons located below the home icon.

4) Move through the plane and zoom in/out so you can better see the details of the previous functions.

When there are too many elements in the "Algebra" view, we can either delete them one by one by selecting the three-point icon and then the "Delete" option, or we can reload the GeoGebra Classic page (though this method requires to activate again the "CAS", "Input Bar", and "3D Graphics" views, if those are required by the user).

5) Delete all the elements in the "Algebra" view.

We can create functions defined with several expressions (each expression being valid in a certain point or interval).

6) Create a function defined through two expressions by entering the following text in the Input Bar:

```
> f(x)=If[x<1,x^2,If[x>=1,1/x]]
```

7) Create a function defined through three expressions by entering the following text in the Input Bar:

```
> f(x)=If[x<1,cos(x-1),If[x>=1 && x < 3,1/x,If[x>=3,sqrt(x)-1.4]]]
```

It is possible to draw the regions associated to inequalities using the <, >, <=, and >= characters.

8) Visualize the following inequalitites:

```
> y>2
```

```
> y+2*x<3
```

```
> x^2-1>2*x
```

4 LIMITS

The limit of a function can be computed using both the "Algebra" and "CAS" views with the command "`Limit(Function,Value)`".

1) Compute the following limit:

```
> Limit(e^(2*x),1)
```

2) Additionally, you can draw the function and identify the point of the graphic with the following commands:

```
> f(x)=e^(2*x)
```

```
> A=(1,f(1))
```

Sometimes, the limit does not exist because the limit from above and the limit from below are different. Those limits can be computed using the commands "`LimitAbove(Function,Value)`" and "`LimitBelow(Function,Value)`", respectively.

2) Compute the limit, the limit from above, and the limit from below of the function $f(x)$ by entering the following commands:

```
> f(x)=x/sqrt(1-cos(x))
```

```
> Limit(f,0)
```

```
> LimitAbove(f,0)
```

```
> LimitBelow(f,0)
```

5 DERIVATIVES AND FUNCTION REPRESENTATION

The derivative of a function is computed by means of the command "`Derivative(Function)`".

1) Compute the derivative of the following functions $f(x)$ (and the specific value of the derivative at $x = 1$) by entering the following commands:

```
> f(x)=x^2*cos(x)-3x
```

```
> f'(x)=Derivative(f)
```

(Alternatively, it is possible to simply enter: `f'`)

```
> f'(1)
```

The second derivative can be computed in the same way using as input function the first derivative.

2) Calculate the second derivative of $f(x)$ through the following command:

```
> f''(x)=Derivative(f')
```

(Alternatively, it is possible to simply enter: `f''`)

The maxima and minima of $f(x)$ can be obtained with the command "`Extremum(f)`".

3) Identify the maximum and minimum points of the following function $f(x)$ by entering the following command:

```
> f(x)=x^3-x
> Extremum(f)
```

Inflection points can be calculated with the command "`InflectionPoint(Function)`".

4) Identify the inflection points of the previous function $f(x)$ with this command:

```
> InflectionPoint(f)
```

Asymptotes can be calculated with the command "`Asymptote(Function)`".

5) Compute the asymptotes of the following function $f(x)$ with these commands (zoom out if you do not see the whole trace of the function):

```
> f(x)=x^2/(x-2)
> Asymptote(f)
```

6 FUNCTION APPROXIMATION

Using GeoGebra, it is possible to compute the Taylor polynomial of degree n of a function centered at the point $x = c$ using the command "`TaylorPolynomial(function,c,n)`".

1) Calculate the Taylor polynomial of degrees 2, 4, and 6 of the function $f(x) = cos(x)$ at $x = 0$ using the following commands:

```
> f(x)=cos(x)
> TaylorPolynomial(f, 0, 2)
> TaylorPolynomial(f, 0, 4)
> TaylorPolynomial(f, 0, 6)
```

GeoGebra also allows to create the interpolation polynomial of degree $n - 1$ taking as input n points with the command "`Polynomial({A,B,...,F})`".

2) Calculate the interpolating polynomial that passes through the points $(1,2)$, $(3,3)$, and $(-2,1)$.

```
> A=(1,5)
> B=(3,3)
> C=(-2,1)
> Polynomial({A,B,C})
```

7 INTEGRALS

Indefinite integrals can be solved with the command "`Integral(Function)`", while definite integrals from $x = a$ to $x = b$ can be computed with the command "`Integral(Function,a,b)`".

1) Compute the indefinite integral of the function $f(x) = x \cdot e^{2x}$ with these commands:

```
> f(x)= x*e^(2x)

> Integral(f)
```

2) Compute the definite integral of the function $f(x) = sin(2x - 1)$ from $x = 0$ to $x = 3 \cdot pi/2$ with these commands:

```
> f(x)=sin(2x-1)

> Integral(f,0,3*pi/2)
```

Of course, if we want to compute the (geometric) area enclosed by the function $f(x)$ and the X-axis, we must calculate the integral of the absolute value of the function.

3) Compute the (geometric) area enclosed by the previous function $f(x)$ with the same starting and end points with the following command:

```
> Integral(abs(f),0,3*pi/2)
```

When trying the determine the area enclosed by two functions and the lines $x = a$ and $x = b$, the command that must be used is "`IntegralBetween(Function1,Function2,a,b)`".

4) Compute the area enclosed by the functions $f(x) = \sqrt{x}$ and $g(x) = x^2$ from $x = 0$ to $x = 1$ with this command:

```
> f(x)= sqrt(x)

> g(x)= x^2

> IntegralBetween(f,g,0,1)
```

If we want to determine the area enclosed solely by two functions, we must first determine the points where $f(x) = g(x)$. If we are working with polynomials, we can use the command "`Solve(Function1=Function2)`".

5) Determine the area completely enclosed by the following functions $f(x)$ and $g(x)$:

```
> f(x)=(x+1)^3-2*(x+1)-2
> g(x)=x^2+2*x-1
> Solve(f=g)
> IntegralBetween(f,g,-2,1)
```

Have you obtained the geometric area? No! That was the algebraic area. In order to obtain the geometric area we would need to compute the integral of the absolute value of the difference $f(x) - g(x)$ with this command:

6) Compute the (geometric) area completely enclosed by the following functions $f(x)$ and $g(x)$:

```
> Integral(abs(f-g),-2,1)
```

If we want to obtain the rational number representing the solution, we must perform the calculation using the "CAS" view.

7) Compute the (geometric) area completely enclosed by the previous functions $f(x)$ and $g(x)$ using the "CAS" view and the same command as before.

When the goal is to determine the length of the trace associated to a function (i.e., its arc length) between two points $x = a$ and $x = b$, we need compute the integral of the square root of the expression $(1 + (f'(x))^2)$.

8) Compute the length of the trace of function $f(x) = (x^3/6) + 1/(2x)$ in the interval $[1,3/2]$ both in the "Algebra" and "CAS" views with these commands:

```
> f(x)= (x^3/6)+(1/(2x))

> Integral(sqrt(1+(f')^2),1,3/2)
```

8 3D VISUALIZATION

We can visualize functions of two variables as surfaces through the "3D Graphics" view. You can disable the “Graphics” and “CAS” view for these examples.

1) Activate the "3D Graphics" view and visualize the following functions:

```
> f_1(x,y)=(x-1)^2 + (y-2)^2 -1

> f_2(x,y)=(x-1)^2 + (y-2)

> f_3(x,y)=x^2-2*x*y
```

It is possible to determine the intersection of a surface and a plane by defining both functions and then clicking in the "Intersect Two Surfaces" icon (seventh icon starting from the left hand side).

2) Determine the intersection of the function following function $f(x, y)$:

```
> f(x,y)=(x-1)^2+(y-2)

> 3*x+5*y-20=0
```

It is also possible to define a parametric curve and determine the tangent and normal lines at one of its points.

3) Enter the following commands for creating a parametric curve and determing the tangent and normal lines at a point:

```
> f(t)=Curve((cos(t),sin(t),0.1*t),t,0,10*pi)

> f(4*pi)

> g(t)=f(4*pi)+f'(4*pi)*t

> h(t)=f(4*pi)+f''(4*pi)*t
```

GeoGebra also allows to visualize the tangent planes to surfaces.

4) Create a sphere centered at the origin of coordinates and visualize the tangent plane at one of its points with the following commands:

```
> x^2+y^2+z^2=4

> A=(0,sqrt(2),sqrt(2))
```

Create a line that passes through point A and the origin (third icon from the left hand side).

Create a tangent plane using point A and the previous line (eighth icon from the left hand side).